

En cette époque de crise de l'énergie, voilà un programme qui vous permettra de trouver, entre autres, la position optimale de votre capteur solaire préféré. (Il est à noter que si votre micro est alimenté par énergie solaire, vous vous trouverez devant un cercle particulièrement vicieux !)

**L**e programme permet de déterminer la position d'un capteur solaire en fonction du lieu, de la date et de l'heure.

Cela permettra à l'astronome amateur de connaître la position exacte du soleil, mais servira également au jardinier soucieux d'obtenir le meilleur ensoleillement de ses serres ou châssis. Cette information pourra également concerner l'architecte pour l'orientation d'une façade d'habitation ; elle ravira certainement l'amateur de cadrans solaires qui pourra enfin « remettre ses pendules à l'heure », sans pour cela attendre d'hypothétiques jours d'ensoleillement maximal.

Exemple : quelle est la position optimale pour un capteur solaire le 23 juin 1983 à 14 h 15 (heure légale) à Marseille (43°25' Nord et 5°13' Est) ?

Réponse (en degrés et dixièmes de degrés) :

Inclinaison : 21,03°

Azimut : 21,83°.


FRAPPE	AFFICHAGE SUR L'ÉCRAN
RUN	Programme position capteurs Année bissextile Si oui taper 1 ?
0 Return	Jour passage heure d'été ?
1 Return	Mois ?
4 Return	Jour passage heure d'hiver ?
1 Return	Mois ?
10 Return	Longitude degrés ?
5 Return	Minutes ?
13 Return	Longitude Ouest ? Si oui taper 1
0 Return	Latitude degrés ?
43 Return	Minutes ?
25 Return	Mois orientation ?
6 Return	Jour orientation ?
23 Return	Heure T.U. sinon heure légale ? Si oui taper 1
1 Return	Heure orientation ?
14 Return	Minutes
15 Return	Inclinaison 21.03° Azimut 21.83° Nouveau mois ?

L'intérêt de ce programme est qu'il évite de longs et fastidieux calculs, et qu'il élimine la possibilité de se tromper dans cette suite de calculs.

La variable M2 du programme étant la hauteur du soleil, on peut, grâce à une modification mineure (impression de M2 au lieu de K (1)) obtenir les deux paramètres qui donnent la position du soleil : hauteur-azimut.

# Tournesol

## ORGANIGRAMME « POSITION DU CAPTEUR »



```

100 PI = 3.14159
110 PRINT "PROGRAMME TOURNESOL"
120 DIM M(26),Y(26),J(3),L(4),K(2),H(3)
130 FOR I = 1 TO 26
140 READ M(I)
150 NEXT I
160 DATA 0,1,2,3,4,5,6,7,8,9,10,11,12
170 DATA 0,1,2,3,4,5,6,7,8,9,10,11,12
180 FOR I = 1 TO 26
190 READ Y(I)

200 NEXT I
210 DATA 0,31,60,91,121,152,182,213
220 DATA 244,274,305,335,366
230 DATA 0,31,59,90,120,151,181,212
240 DATA 243,273,304,334,365
250 PRINT "ANNE BISSEXTILE"
260 INPUT "SI OUI TAPER 1";A
270 INPUT "JOUR PASSAGE HEURE ETE ";J(2)
280 INPUT "MOIS ";M
290 IF A = 1 GOTO 330

300 M = M + 13
310 V = J(1) + Y(M)
320 GOTO 340
330 V = J(1) + Y(M)
340 INPUT "JOUR DE PASAGE HEURE HIV ";J(2)
350 INPUT "MOIS ";M
360 IF A = 1 GOTO 410
370 M = M + 13
380 W = J(2) + Y(M)
390 Z = 80

400 GOTO 430
410 W = J(2) + Y(M)
420 Z = 81
430 INPUT "LONGITUDE DEGRES ";L(1)
440 INPUT "MINUTES ";L(2)
450 PRINT "LONGITUDE OUEST "
460 INPUT "SI OUI TAPER 1 ";B
470 IF B < > 1 THEN GOTO 480
475 L = - L
480 INPUT "LATITUDE DEGRES ";L(3)
490 INPUT "MINUTES ";L(4)

500 INPUT "MOIS D'ORIENTATION ";M
510 INPUT "JOUR D'ORIENTATION ";J(3)
520 IF A = 1 GOTO 560
530 M = M + 13
540 N = J(3) + Y(M)
550 GOTO 570
560 N = J(3) + Y(M)
570 PRINT "HEURE TU SINON HEURE LEGALE"
580 INPUT "SI OUI TAPER 1";D
590 INPUT "HEURE ORIENTATION ";H(1)

600 INPUT "MINUTE ";H(2)
610 IF D = 1 GOTO 660
620 PRINT "CALCUL EN HEURE LEGALE"
630 IF N < V GOTO 650
640 IF N > = W GOTO 650
645 H(1) = H(1) - 1
650 H(1) = H(1) - 1
660 N1 = N - Z
670 D1 = 0.4 * SIN ((2 * PI * N1) / 365)
680 D2 = ATN (D1 / SQR (1 - D1 ^ 2))
690 L = L(1) + L(2) / 60
700 A = - 0.25888E + 1

```

```

701 B = - 0.66626E + 0
702 C = 0.19372E - 1
703 D = - 0.4962E - 3
704 P = 0.96541E - 5
710 F = - 0.1057E - 6
711 U = 0.65537E - 9
712 H = - 0.23814E - 11
713 I = 0.50472E - 14
714 J = - 0.5796E - 17
720 K = 0.27946E - 20
730 C = A + B * N + C * (N ^ 2) + D * (N ^ 3) + P * (
      N ^ 4) + F * (N ^ 5) + U * (N ^ 6) + H * (N ^ 7) +
      I * (N ^ 8) + J * (N ^ 9)
740 C = C + K * (N ^ 10)
750 C = C + (4 * L)
760 H(2) = H(2) + C
770 IF H(2) > 60 GOTO 820
780 IF H(2) > 0 GOTO 840
790 H(2) = 60 + H(2)

800 H(1) = H(1) - 1
810 GOTO 840
820 H(2) = H(2) - 60
830 H(1) = H(1) + 1
840 H(3) = H(1) + H(2) / 60
850 P1 = L(3) + L(4) / 60
860 P = P1 * 2 * PI / 360
870 E = SIN (P) * SIN (D2)
880 IF H(3) > = 12 GOTO 890
885 H(3) = - (12 - H(3))
886 GOTO 900
890 H(3) = H(3) - 12

900 AH = (2 * PI * H(3)) / 24
910 M = E + COS (D2) * COS (P) * COS (AH)
920 M1 = ATN (M / SQR (1 - (M ^ 2)))
930 M2 = (M1 * 360) / (2 * PI)
940 K(1) = 90 - M2
950 K(1) = INT (K(1) * 100) / 100
960 Q = (COS (D2) * SIN (AH)) / COS (M1)
970 K(2) = ATN (Q / SQR (1 - (Q ^ 2)))
980 K(2) = (360 * K(2)) / (2 * PI)
990 K(2) = INT (K(2) * 100) / 100

1000 PRINT
1010 PRINT
1020 PRINT
1030 PRINT "INCLINAISON ";K(1);" DEG"
1040 PRINT
1050 PRINT
1060 PRINT "      AZIMUT ";K(2);" DEG"
1070 PRINT
1080 PRINT
1090 PRINT

1100 PRINT "NOUVEAU MOIS ?"
1110 INPUT "SI OUI TAPER 1 ";F
1120 IF F = 1 GOTO 500
1130 PRINT "CALCUL MEME MOIS"
1140 PRINT "NOUVEAU JOUR"
1150 INPUT "SI OUI TAPER 1";F
1160 IF F = 1 GOTO 510
1170 PRINT "NOUVELLE HEURE ?"
1180 INPUT "SI OUI TAPER 1";F
1190 IF F = 1 GOTO 570
1200 END

```

*pour Commodore  
4016  
Transposition  
très facile*